


BLE M

COMET

MYRESNAK



ICL

Listning af programmerne som er benyttet på brugervejledningens omslag :

MYREBEN(R)
HØJRE(R*30)
FREM(20)
HØJRE(R*135)
FREM(18)
HØJRE(R*105)
BAK(8)
FLYV
FREM(22)
KRAVL
NORD
SLUT

MYREDELSIDE
FREM(256)
STRAM
HØJRE(75)
FREM(535)
HØJRE(135)
FREM(354)
HØJRE(45)
FREM(354)
HØJRE(105)
FREM(262)
HØJRE(75)
FREM(500)
SLUT

MYREKROP
VENSTRE(90)
FREM(20)
HØJRE(135)
GENTAG(3)
FREM(14)
VENSTRE(45)
MYREBEN(-1)
HØJRE(45)
HERTIL
FREM(14)
NORD
GENTAG(3)
BAK(10)
MYREBEN(1)
HERTIL
BAK(9)
HJEM
SLUT

MYREFORSIDE
HENT(MYREKROP)
HENT(MYREBEN)
HENT(MYREDELSIDE)
MYREKROP
VENSTRE(150)
FREM(200)
HØJRE(60)
SNOR
FREM(250)
HØJRE(135)
FREM(250)
STRAM
SNOR
FREM(100)
HØJRE(45)
FREM(350)
HØJRE(105)
MYREDELSIDE
SLUT

Under forudsætning af at MYREFORSIDE ligger i afbejdsområdet, vil ordren MYREFORSIDE starte programudførelsen. De tre underprogrammer vil automatisk hentes fra disketten, og 'forsiden' vil derefter blive tegnet.

INDHOLDSFORTEGNELSE

1.	Hvad er myresnak ?	1.1
2.	Beskrivelse af systemet myresnak	2.1
2.1	Dataskærm	2.1
2.2	Matrixprinter	2.2
2.3	Systemdisketten	2.2
3.	Installation af myresnak	3.1
3.1	Fremstilling af sikkerhedskopi	3.1
3.2	Fremstilling af arbejdskopi	3.1
4.	Opstart af myresnak	4.1
4.1	Udskrivning af diskettekataloget på skærmen	4.1
4.2	Udskrivning af diskettekataloget på linseskriveren	4.1
4.3	Sletning af filer på disketten	4.2
4.4	Omdøbning af filer på disketten	4.2
5.	Ordreerne til myresnak	5.1
5.1	Beskrivelse af ordreerne alfabetisk ordnet	5.1
5.2	Oversigt over ordreerne til myresnak	5.10
6.	Programmer	6.1
6.1	Programnavn	6.1
6.2	Variabler	6.1
6.3	Programdel	6.1
6.4	Programkald	6.2
6.5	Udtryk	6.2
6.6	Tal	6.2
6.7	Regnetegn	6.2
6.8	Sløjfer	6.2
6.9	Betingede ordrer	6.3
6.10	Udsagn	6.3
6.11	Logisk operator	6.3
6.12	Programafbrydelse	6.4
7.	Indskrivning af programmer	7.1
8.	Variabelbegrebet i myresnak	8.1
8.1	Globale variabler	8.1
8.2	Lokale variabler	8.2

9.	Lidt om myresnak	9.1
9.1	Specifikationer	9.1
9.2	Fortolkningen	9.2
10.	Efterskrift	10.1

1. Hvad er MYRESNAK ?

Myresnak er en fortolker, der bygger på den delmængde af LOGO-projectet, som betegnes TURTLETALK.

I myresnak styrer eleven ved ordrer en 'myre', som bevæger sig rundt på fingrafikskærmen. Ved opstarten af myresnak kender myren kun ganske få og simple ordre (f.ex. frem, bak, højre, venstre, flyv o.s.v), men eleven har mulighed for selv at definere nye ordrer, som myren derefter husker og udfører på kommando.

Myresnak er altså et stærkt elev/maskin interaktivt programmeringsprog.

Pædagogisk set har myresnakdialogen mange fordele. Hver ordre giver øjeblikkelig respons på skærmen. Gør myren ikke, hvad eleven forventer, kan fejlen straks rettes. Eleven lærer at algoritmisere en række af ordrer, når der arbejdes med programmer. Desuden animerer sproget til at opdele store programmer i mindre delprogrammer. Der er mulighed for store kreative udfoldelser på skærmen.

Fortolkeren foretager ingen syntaks-check af ordrer, der indtastes direkte eller gives i programmer. Kan den ikke genkende ordren eller eventuelt programnavnet, springer den det over og giver et bip i højtaleren (hvis anlægget har en sådan indbygget). Dette medfører, at brugeren tvinges til at arbejde omhyggeligt og grundigt med sine programstrukturer, hvorved gode arbejdsvaner oparbejdes.

Myresnak henvender sig til undervisningssektoren. I folkeskolen vil den kunne bruges fra ca. 3.-4. klassetrin til 10. Desuden egner den sig udmærket til brug i gymnasiet, HF, EFG og mange andre steder.

Denne myresnak fortolker følger nøje de specifikationer, som er givet af Viggo Sadolin ved Danmarks Lærerhøjskole, således at undervisningsmateriale herfra til myresnak umiddelbart kan bruges. På visse punkter er denne fortolker udvidet. Ordre-sættet er forøget. Antallet af programmer fortolkeren, kan behandle samt antallet af kommando-niveauer er også stærkt udvidet. Desuden benytter denne fortolker en program-editor, som er meget nem at betjene.

Der er udkommet en elevtekst til brug i folkeskolens matematik-undervisning :

MYRESNAK - Matematik 6.-10. klassetrin.
Viggo Sadolin
Teknisk Forlag, September 1984.

2. Beskrivelse af systemet til myresnak.

I forbindelse med myresnak skal man anvende en COMET basis-model udbygget med følgende:

- Grafikmodul MPS-24
- Matrixprinter med grafikprint.

2.1 Dataskærm.

Under myresnak skiftes mellem 2 billeder på dataskærmen,

- et videobillede (tekstsider)
- et grafikbillede (grafikskærm nr. 1)

På begge skærme findes nedest to linier (indrammede), hvor ordrene skrives. På øverste linie vises den sidst udførte ordre, mens nederste linie benyttes til indtastning af ordrer, der skal eksekveres.

2.1.1 Videobilledet.

Videobilledet vises under opstart og bruges, når der arbejdes med myreprogrammer og pladelageret. Der skiftes automatisk til video, når de respektive systemordrer eksekveres.

Der er altså ingen specielle ordrer til kald af dette billede, men f.eks. ordren KATALOG vil foretage skiftet.

I anden linie på højre halvdel af videobillede oplyses, om fortolkeren arbejder med LOKALE eller GLOBALE variable (mere herom i kap. 8).

2.1.2 Grafikbilledet.

Grafikbilledets tegneområde består af et indrammet tegnefelt, der er 505 punkter bredt og 465 punkter højt.

Ved opstart samt ved ordrene START, HJEM og OPSTART anbringes myren i midtpunktet af dette felt, altså i punktet med koordinaterne (252,232).

Alle myreordrer udføres på grafikskærmen.

Grafikbilledet kaldes ved ordrene 'TEGNING' og 'RENS', eller når et programnavn skrives.

2.2 Matrixprinter.

Under fremstillingen af arbejdskopier af originaldisketten med myresnak (se kap. 3.2) tilpasses fortolkeren automatisk den rigtige printertype.

Ordren TRYK vil kopiere grafikbilledet over på matrixprinterens uanset hvilket skærbillede, der vises.

Udskriften kan afbrydes ved tryk på en vilkårlig tast.

Myresnak fortolkeren kan udmærket køre uden printerens tilsluttet, men i sagens natur vil ordren TRYK så ingen mening have.

2.3 Systemdisketten.

På systemdisketten ligger følgende programmer :

- MYRESNAK.COM
- MYRE82.COM
- MYRE83.COM
- MYRE92.COM
- MYREDOKO.PAS
- MYREDOK1.PAS
- MYREDOK2.PAS
- MYREDOK3.PAS

HUSK: Myresnak kan ikke køres fra systemdisketten. Der skal laves en arbejdskopi (se kap. 3.2).

3. Installation af myresnak.

Fremstil straks en sikkerhedskopi af originaldisketten.

BEMÆRK: Kun originaldisketten kan ombyttes, hvis der konstateres fejl i leverancen.

Konstateres fejl under brugen af myresnak laves en ny arbejdskopi.

3.1 Fremstilling af sikkerhedskopi.

1. Indsæt originaldisketten i pladelager A.
2. Indsæt en formatteret diskette i pladelager B.
3. Tryk på 'RESET' knappen.
4. Tryk på '1'-tasten.
Hvis der ikke vises 'A>' på skærmen nedtrykkes funktions-tasten øverst til venstre og pkt. 4 gentages.
5. Indtast 'PIP B:=*.*ÆVA' og tryk på 'RETURN'.
Originaldisketten kopieres over på disketten i lager B.

3.2 Fremstilling af arbejdskopi.

1. Udfør punkterne 1-4 under kap. 3.1.
2. Indtast 'MYRESNAK' (tryk 'RETURN').
3. Besvar spørgsmålet om printertype.
En arbejdskopi lagres nu på disketten i pladelager B.
4. Ønskes flere arbejdskopier, udskiftes disketten i B med en formatteret diskette, og spørgsmålet besvares med 'J'.

4. Opstart af myresnak.

Nedenstående punkter skal nøje følges ved opstart af myresnak-fortolkeren.

1. Tænd for COMETen og de evt. tilsluttede enheder.
2. Indsæt disketten i det pladelager, hvori opstart normalt foretages (A på enbruger og B på flerbruger systemer).
3. Tryk på '1'-tasten.
Hvis der ikke vises 'A>' på skærmen nedtrykkes funktions-tasten øverst til venstre og pkt. 3 gentages.
4. Indtast 'MYRESNAK' og tryk på 'RETURN'.
5. Myresnak er nu klar til at modtage ordrer.

4.1 Udskrivning af diskettekataloget på skærmen.

A. Inden opstart af systemet.

1. Udfør pkt. 1-3 under opstart (kap. 4.)
2. Indtast 'DIR'.
Kataloget vises på skærmen.
3. Udfør pkt. 4-5 under opstart.

B. Mens myresnak er i brug.

1. Indtast 'STOP'.
Myresnak-fortolkeren standser. Husk at gemme eventuelle programmer inden ordren 'STOP', ellers mistes de.
2. Indtast 'DIR'.
3. Udfør pkt. 4-5 under opstart.

4.2 Udskrivning af diskettekataloget på linieskriveren.

A. Inden opstart af systemet.

1. Udfør pkt. 1-3 under opstart (kap. 4.)
2. Hold 'CONTROL' nede, mens der trykkes een gang på 'P'.
3. Indtast 'DIR'.
Kataloget udskrives på skriveren samtidig med at det vises på skærmen.
4. Udfør pkt. 4-5 under opstart.

B. Mens myresnak er i brug.

1. Indtast 'STOP'.
Myresnak-fortolkeren standser. Husk at gemme eventuelle programmer inden ordren 'STOP', ellers mistes de.
2. Hold 'CONTROL' nede, mens der trykkes een gang på 'P'.
3. Indtast 'DIR'.
4. Udfør pkt. 4-5 under opstart.

4.3 Sletning af filer på disketten.

A. Inden opstart af systemet.

1. Udfør pkt. 1-3 under opstart (kap. 4.).
2. Indtast

ERA <programnavn>.MYR

og tryk RETURN (Navnet er max. tegn lang).
Ønskes alle myreprogrammer slettet, gives kommandoen

ERA *.MYR

3. Udfør pkt. 4-5 under opstart (kap. 4.).

B. Mens myresnak er i brug.

1. Indtast STOP.
Myresnak-fortolkeren standser. Husk at gemme eventuelle programmer inden ordren STOP, ellers mistes de.
2. Indtast

ERA <programnavn>.MYR

og tryk RETURN (Navnet er max. tegn lang).
Ønskes alle myreprogrammer slettet, gives kommandoen

ERA *.MYR

3. Udfør pkt. 4-5 under opstart (kap. 4.).

4.4 Omdøbning af filer på disketten.

Omdøbning af filnavne på disketten foretages på følgende måde:

1. Opstart myresnak (se kap. 4)
2. Indtast HENT(<programnavn>).
3. Indtast RET og skriv programnavnet.
4. Brug editoren til at rette programnavnet.
5. Indtast GEM og skriv programnavnet.
6. Indtast STOP og maskinen svarer med 'A'.
7. Slet den gamle fil (se kap. 4.3).

5. Ordreerne til myresnak.

Alle ordreer kan frit benyttes i alle programmer, selv systemordreerne. Dog kan brugen af ordreer, som ikke styrer selve myren, give mærkelige virkninger. Kørende programmer kan altid standses ved tryk på en vilkårlig tast.

5.1 Beskrivelse af ordreerne alfabetisk ordnet.

5.1.1 BAK (forkortet BK)

Syntaksen er BAK(<tal>), hvor <tal> er et heltal. Myren bevæger sig det angivne antal punkter tilbage. Udelades tallet, tolkes det som BAK(0), og myren bevæges ikke.

Tallet kan være negativt, hvilket bevirker, at myren bevæger sig <tal> punkter frem.

Bruges BAK som programordre kan <tal> erstattes med en variabel eller et udtryk (se kap. 6.5).

Eks. BAK(132) Myren bevæges 132 punkter tilbage.
 BAK(L/2) Myren bevæger sig baglæns det halve antal punkter angivet i variabel L.

5.1.2 ERYM

Myren fjernes fra skærmen (gør sig usynlig).
Myren gøres synlig igen ved ordreerne MYRE eller OPSTART.

5.1.3 FLYV

Myren løftes op fra sandet, dvs. myren trækker ikke spor efter sig.

Myren giver igen spor i sandet ved ordrene KRAVL eller OPSTART.

5.1.4 FREM (forkortet FR)

Syntaksen er FREM(<tal>), hvor <tal> er et heltal.

Myren bevæger sig det angivne antal punkter frem. Angives intet tal vil myren ikke bevæge sig.

Tallet kan godt være negativt, hvilket bevirker, at myren bevæger sig tilbage.

Som programordre kan <tal> erstattes af en variabel eller et udtryk (se kap. 6.5).

Eks. FREM(34) Myren bevæges 34 punkter frem.
 FR(L) Myren bevæges frem det antal punkter, som variabelen L angiver.

5.1.5 GEM

Ordren gives, når man ønsker at gemme et program fra arbejdsområdet på pladelageret.

Der skiftes automatisk til videobillede og kataloget vises samtidig med, at der spørges efter programnavn. Når dette er indtastet, gennemses kataloget, og findes programmet, kopieres det over på disketten under samme navn. Findes programmet ikke, gives en fejlmeddelelse.

Se iøvrigt under programnavn i kap. 6.1.

5.1.6 GENTAG

Denne ordre er en programordre. Gives den på kommandoniveau (niveau 0) overspringes den af fortolkeren.

Ordren er startordren for en sløjfestruktur, som afsluttes med ordren HERTIL (se den nærmere beskrivelse i kap. 6.8).

Orden kan bruges på følgende 3 måder:

1. GENTAG uden parameter gentager det efterfølgende i det uendelige.
2. GENTAG(<tal>) , hvor <tal> er et heltal større end nul. Sløjfen gentages <tal> gange.
3. GENTAG(<tal>) , hvor <tal> er nul eller et negativt heltal. Hele sløjfen overspringes og sætningen efter ordren HERTIL udføres.

5.1.7 GLEM

Ordren benyttes til at slette et program i arbejdsområdet.

Fortolkeren skifter automatisk til videobillede, viser kataloget og spørger om programnavn.

Hvis programnavnet ikke findes gives en fejlmeddelelse, ellers slettes programmet i arbejdsområdet, og pladsen kan nu benyttes til et nyt program.

5.1.8 GLOBAL

Alle variable gøres globale, hvilket bevirker, at alle myreprogrammer har adgang til det samme variabelsæt.

På videoskærmens anden linie vises, at fortolkeren arbejder globalt.

5.1.9 GRØN

Myren sætter spor på skærmen med grøn farve.

5.1.10 HDREJ (forkortet HD)

Funktionsbeskrivelse se under HØJRE (kap. 5.1.17).

5.1.11 HENT

Ordren benyttes, når et myreprogram skal hentes fra disketten og indlæses i arbejdsområdet.

Fortolkeren skifter automatisk til videobillede og spørger om programnavn. Når dette er indtastet, indlæser fortolkeren programmet og viser kataloget som kontrol.

Ordren hent kan i programmer benyttes med parameter således:

HENT(<programnavn>)

Fortolkeren henter programmet fra disketten og indlæser det i arbejdsområdet. Der skiftes ikke til videosiden, kataloget vises ikke og evt. fejlmeddelelser undertrykkes.

Eks.	TREKANT(L)	KVADRAT(L)	HUS(L)
	GENTAG(3)	GENTAG(4)	HENT(TREKANT)
	FR(L)	FR(L)	HENT(KVADRAT)
	HØ(120)	HØ(90)	KVADRAT(L)
	HERTIL	HERTIL	FR(L)
	SLUT	SLUT	HØ(30)
			TREKANT(L)
			SLUT

Afgives ordren HUS(50) udfører fortolkeren følgende:
Programmet TREKANT hentes fra disketten til arbejdsområdet.
- KVADRAT - - - -

Findes TREKANT eller KVADRAT allerede i arbejdsområdet, indlæses der intet program fra disketten, men det gamle program kaldes i stedet af programmet HUS.

Et kvadrat med sidelængde 50 tegnes, myren flyttes 50 punkter nordpå, drejes 30 grader mod højre og en trekant med sidelængden 50 tegnes.

5.1.12 HERTIL

Ordren HERTIL afslutter en sløjfestructur (se kap. 6.8).

Ordren er en programordre. Hvis den bruges på kommandoniveau (niveau 0) overspringer fortolkeren den.

5.1.13 HJEM

Ordren HJEM udfører følgende:

- Myren føres til startpositionen midt på skærmen.
- Myren drejes, så den peger mod nord.
- Hvis myren 'kravlede' trækkes et spor til start.
- Hvis myren var 'usynlig', er den det stadig.

5.1.14 HUSK

HUSK aktiverer program-editoren. Der skiftes til videobillede, og markøren stilles i øverste venstre hjørne af indskrivningsfeltet.

Om brugen af editoren se kap. 7.

5.1.15 HVID

Samme funktion som GRØN (se kap. 5.1.9).

5.1.16 HVIS

HVIS er en betinget ordre, som kun benyttes i programmer. Bruges den på kommandoniveau (niveau 0), overspringer fortolkeren den.

Syntaksen er følgende :

HVIS(<udsagn>><ordre> eller HVIS(<udsagn>><programnavn>

En mere udførlig beskrivelse gives i kap. 6.9.

5.1.17 HØJRE (forkortet HØ)

Syntaksen for ordren er : HØJRE(<tal>).

<tal> er et heltal, som angiver hvor mange grader myren skal drejes højre om. Er <tal> negativt, vil myren dreje venstre om.

Som programordre kan <tal> erstattes af en variabel eller et udtryk (se kap. 6.5).

Eks.	HØJRE(54)	Myren drejes 54 grader højre om.
	HØJRE(V)	Myren drejes det antal grader højre om, som tallet i variabel V angiver.

5.1.18 KATALOG (forkortet KT)

Videobilledet kaldes automatisk, og en oversigt over programmerne i arbejdsområdet vises.

5.1.19 KRAVL

Myren 'kravler', dvs. der tegnes et spor på grafikskærmen.

5.1.20 LIST

Videobilledet kaldes automatisk, og fortolkeren spørger om programnavn. Når dette er indtastet, vises en listning af programmet.

Hvis programmet ikke befinder sig i arbejdsområdet, udskrives en fejlmeddelelse.

5.1.21 LOKAL

Fortolkeren behandler alle variable som lokale variable, dvs. at hver gang der foretages et programkald, oprettes et variabelsæt specielt for dette program. Dette gælder også, hvis et program kalder sig selv (rekursive programkald).

Fortolkeren kan kun behandle 50 forskellige variabelsæt, så det er programmørens ansvar, specielt ved rekursive programkald, at holde sig indenfor maksimalt 50 kald, før tilbageviklingen påbegyndes.

På videoskærmens anden linie vises, at fortolkeren arbejder med lokale variable.

Se desuden beskrivelsen i kap. 8.

5.1.22 MYRE

Myren vises på grafikskærmen.

5.1.23 NUL

Alle variabler nulstilles, uanset om fortolkeren behandler lokale eller globale variabler.

Som programordre skal NUL benyttes med en vis forsigtighed. Variablerne bliver øjeblikkelig nulstillet også dem, der gælder for programmet, der er under udførsel.

Eks. Følgende program ligger i arbejdsområdet.

```
SNYD(L,V)
NUL
FREM(L)
VENSTRE(V+5)
SLUT
```

Kommandoen SNYD(10,50) tillægger variabelen L værdien

10 og variabelen V værdien 50.

L og V får værdien nul. Myren flyttes 0 punkter frem og venstredrejes $0+5 = 5$ grader.

Altså uanset hvilke værdier L og V tillægges, udfører programmet kun det samme som ordren VENSTRE(5).

5.1.24 OPSTART

Ordren OPSTART udfører følgende operationer :

- Myren føres til startpositionen midt på skærmen (START).
- Myren stilles, så den peger opad (NORD).
- Myren gøres 'synlig' (MYRE).
- Myren sætter spor på skærmen (KRAVL).
- Myren sætter spor (GRØN).
- Alle variabler nulstilles (NUL).
- Alle variabler er globale (GLOBAL).
- Snoren fæstnes til denne position (SNOR).
- Niveau-tælleren stilles til 0.

Ved denne opstart startstilles alt, dog vil alle programmerne i arbejdsområdet være bevaret.

Ordren kan benyttes, hvis fortolkeren mod al forventning ikke arbejder, som den skal.

5.1.25 RENS

Følgende operationer udføres ved ordren RENS :

- Der skiftes til grafikskærm (TEGNING).
- Tegninger på grafikskærmen fjernes.
- Myren stilles i startstilling (START).
- Myren stilles, så den peger opad (NORD).
- Alle variabler nulstilles (NUL).

5.1.26 RET

Fortolkeren skifter automatisk til videoskærmen og spørger om programnavn. Når dette er indtastet, listes programmet, og markøren stilles i øverste venstre hjørne i programfeltet.

Editoren aktiveres, og rettelsen kan foretages som forklaret i kap. 7 under indskrivning af programmer.

5.1.27 SKRIV

Ordren SKRIV giver mulighed for udskrift af programmer på linseskriveren.

Fortolkeren skifter automatisk til videoskærmen og spørger om programnavn. Når dette er indtastet, listes programmet på

skærmen og udskrives samtidigt på den tilsluttede matrixprinter.

5.1.28 SLET

Samme ordre som GLEM (se kap. 5.1.7).

5.1.29 SLUT

Ordren SLUT er den ordre, som afslutter et program. Den er samtidig et signal til program-editoren om at afslutte editeringen.

Hvis ordren afgives på kommandoniveauet (niveau 0), tolkes den ikke.

5.1.30 SNOR

Når ordren SNOR afgives, husker fortolkeren koordinaterne til det punkt, hvor myren befinder sig.

Hver gang ordren STRAM afgives, vil der blive trukket en linie fra dette punkt til det sted, hvor myren nu befinder sig.

Først når ordren SNOR igen afgives, vil fortolkeren udskifte koordinatsættet med koordinaterne til det punkt, hvor myren nu står.

Eks. Følgende program ligger i arbejdsområdet :

```
VIFTE
SNOR
GENTAG(12)
FLYV
FREM(50)
HØJRE(30)
KRAVL
STRAM
HERTIL
SLUT
```

Programmet vil tegne alle diagonalerne til startpunktet i en tolvkant samt tegne de to sidestykker, der udgår fra startpunktet.

5.1.31 SORT

Myren tegner med 'sort' farve på skærmen, dvs. alle tændte punkter slukkes.

Ordren bruges til af slette og rette i tegningen.

5.1.32 START

Ved ordren start udfører fortolkeren følgende :

- Myren føres til startstillingen midt på skærmen uden at sætte spor.
- Myren stiller sig i retningen pegende opad.

5.1.33 STOP

STOP ordren afgivet på kommandoniveau (niveau 0) standser fortolkeren og overgiver kommandoen til CP/M.

Man har derefter mulighed for at se, hvad der ligger på disketten med kommandoen DIR, eller slette programfiler med kommandoen ERA <programnavn>.MYR (se kap. 4.3).

Benyttes STOP i et program, standser dette. Niveautallereren sættes til 0, altså kommandoniveauet.

BEMÆRK: Når myresnak fortolkeren standser, mistes alle programmer i arbejdsområdet. Husk derfor at gemme dem på disketten inden STOP ordren afgives.

5.1.34 TEGNING (forkortet TG)

Fortolkeren skifter til grafikbilledet uden at slette det, der befinder sig her. Myrens attributter ændres ikke.

5.1.35 TRYK

En kopi af grafikskærmen kopieres på matrixprinterens uanset det valgte skærbillede.

Udskrivningen kan afbrydes ved tryk på en vilkårlig tast.

5.1.36 UDHOP

UDHOP er en programordre, som bruges til udhop fra en sløjfe. Ordren har kun mening, hvis den befinder sig inden i sløjfen, dvs. mellem ordrene GENTAG og HERTIL. Bruges den andre steder, vil virkningerne utvivlsomt virke overraskende..

Når fortolkeren udfører UDHOP ordren, foretager den et spring til sætningen umiddelbart efter ordren HERTIL, hvorefter den tolker denne sætning.

5.1.37 VDREJ (forkortet VD)

Ordren har syntaksen : VDREJ(<tal>) eller VD(<tal>)
og er identisk med ordren VENSTRE (se kap. 5.1.38).

5.1.38 VENSTRE (forkortet VE)

Ordren har syntaksen : VENSTRE(<tal>) eller VE(<tal>)

<tal> er et heltal og angiver, hvor mange grader myren
skal dreje sig venstre om.

<tal> kan være negativ, hvilket bevirker, at myren drejer
højre om.

Som programordre kan <tal> erstattes af en variabel eller
et udtryk (se kap. 6.5).

Eks. VENSTRE(45) Myren drejes 45 grader venstre om.
 VENSTRE(V) Myren drejes det antal grader venstre
 om, som tallet i variabel V angiver.

5.2 Oversigt over ordrerne til myresnak.

Ordrerne kan inddeles i tre dele. Et sæt ordrer, der styrer myren. Et sæt, der bruges i programmerne og et sæt, der bruges til styring af de tilsluttede enheder samt styring af variablerne.

5.2.1 Myreordrer.

Ordre	Forkortet	Funktion
HØJRE(<tal>)	HØ(<tal>)	Myren drejer <tal> grader mod højre.
HDREJ(<tal>)	HD(<tal>)	- - - - -
VENSTRE(<tal>)	VE(<tal>)	Myren drejer <tal> grader mod venstre.
VDREJ(<tal>)	VD(<tal>)	- - - - -
FREM(<tal>)	FR(<tal>)	Myren går <tal> punkter frem.
BAK(<tal>)	BK(<tal>)	Myren bakker <tal> punkter.
MYRE		Myren vises på skærmen.
ERYM		Myren fjernes fra skærmen.
FLYV		Myren kan bevæges uden at sætte spor.
KRAVL		Myren sætter spor på skærmen.
SNOR		Myren husker hvor den er på skærmen.
STRAM		Myren trækker en linie fra sin nuværende position til positionen, hvor ordren SNOR blev givet (Snoren strammes).
NORD		Myren drejes mod nord (lodret op).
START		Myren 'flyver' til startstillingen.
HJEM		Myren trækker et spor tilbage til startstillingen.
GRØN eller HVID		Myren tegner grønt spor på skærmen.
SORT		Myren tegner sort (sletter sit spor).

5.2.2 Programordrer.

Ordre	Funktion.
GENTAG	Gentag det følgende i det uendelige.
GENTAG(<tal>)	- - - - - <tal> gange. (Hvis <tal> <= 0 udføres sætningen efter HERTIL.
HERTIL	Afslutning af gentage-sløjfen.
HVIS(<udsagn><ordre>)	Hvis <udsagn> er sandt udføres <ordre> ellers overspringes sætningen.
UDHOP	Sætningen efter hertil udføres.
SLUT	Afslutter programmet.

5.2.3 Systemordrer.

Ordre	Forkortet	Funktion
TEGNING	TG	Der skiftes til tegnesiden.
RENS		Skærmen renses, og myren anbringes i startstillingen.
KATALOG	KT	Alle programnavnene vises.
HUSK		Nyt program kan indtastes.
LIST		Udskrift af program på skærmen.
SKRIV		Udskrift af program på lineskriveren.
TRYK		Kopierer tegnesiden på lineskriveren.
RET		Man kan rette i et eksisterende program.
GLEM eller SLET		Sletning af program i arbejdsområdet.
NUL		Nulstiller alle variabler.
HENT		Henter program på disketten.
GEM		Lagrer program på disketten.
LOKAL		Myresnak arbejder med lokale variabler.
GLOBAL		- - - globale - -
STOP		Fortolkeren stopper (Husk at alle programmer i arbejdsområdet mistes, hvis de ikke er gemt på disketten).
OPSTART		Fortolkeren opstartes påny, dog vil alle programmerne i arbejdsområdet være bevaret.

6. Programmer.

Et program består af

1. <programnavn><variabel>, ..., <variabel>
2. programdel
3. SLUT

Et program kan maksimalt bestå af 20 linier med 30 tegn.

6.1 Programnavn.

Programnavnet kan bestå af et antal bogstaver eller tal (evt. blandet) og kan være lige så langt, som linien tillader.

Programnavnet kan ikke være en af ordrene nævnt i kap. 5.

Hvis programmet skal gemmes på diskette, bør programnavnets længde ikke overstige 8 tegn, da fortolkeren (af hensyn til CP/M systemet) automatisk afskærer tegn udover det ottende, når der arbejdes på disketten.

Eks. På disketten ligger programmet 'FIRKANTER' under navnet 'FIRKANTE' (8 tegn).

Programmet 'FIRKANTERNE' gemmes.

Programmet 'FIRKANTE' (opr. 'FIRKANTER') overskrives, og efter operationen ligger programmet 'FIRKANTERNE' nu på disketten under navnet 'FIRKANTE'.

Hentes programmet ind i fortolkeren, vises nu i kataloget programmet 'FIRKANTERNE'.

Dvs. at hvis et programnavn overstiger 8 tegn vil diskettekataloget kun vise de første 8.

6.2 Variabler.

En variabel består af eet bogstav (A-Z).

Bruges flere variabler, skilles de ved komma.

Der kan maksimalt benyttes 8 variabler pr. program.

6.3 Programdel.

Programdelen består af

1. Ordrer
2. evt. een sløjfe
3. evt. programkald

Eks.

VINKEL(V)
FREM(200)
VENSTRE(V)
FREM(200)
SLUT

KVADRAT
GENTAG(4)
FREM(100)
HØJRE(90)
HERTIL
SLUT

SPIRAL(Q)
FREM(10)
VENSTRE(Q)
SPIRAL(Q+10)
SLUT

6.4 Programkald.

Programkald består af

<programnavn>(<udtryk>, ..., <udtryk>)

<udtryk> kan udelades.

Det er vigtigt, at antallet af udtryk til programmet svarer til antallet af programmets variable.

6.5 Udtryk.

Et udtryk opfattes som

1. et <tal> eller en <variabel>
2. to <tal> eller <variable> med et <regnetegn> imellem

Eksempler på LOVLIGE udtryk:

2+5 A-3 7*L -Q A/L 5/-B -2*P 5*-A

Eksempler på ULOVLIGE udtryk:

2*5+6 2*A/R A*B*C

6.6 Tal.

Myresnak fortolkeren behandler kun heltal, som kan være positive, negative eller nul. Talområdet går fra -32768 til 32767.

6.7 Regnetegn.

Fortolkeren arbejder kun med '+', '-', '*' og '/'.

6.8 Sløjfer.

Sløjfer består af et antal programordrer eller programkald, som skal gentages et vist antal gange evt. uendelig mange gange. Syntaksen er følgende

1. GENTAG(<udtryk>)
2. ordredel
3. HERTIL

Ad. 1. Hvis <udtryk> udelades, gentages ordredelen i det uendelige.

Ad. 2. Hvis <udtryk> er positivt, udføres ordredelen <udtryk>

antal gange. Sløjfen kan afbrydes, hvis ordren UDHOP bruges evt. i en betinget ordre. Fortolkeren udfører så linien efter HERTIL.

Ad. 3. Hvis <udtryk> er nul eller negativ, overspringes ordredelen, og linien efter HERTIL udføres.

BEMÆRK: Fortolkeren tillader kun een sløjfe pr. program. Ønskes flere sløjfer (evt. inden i hinanden), gøres det ved at kalde et nyt program, der udfører denne sløjfe.

Eks.	SPIRAL(L)	TREKANT(L)
	GENTAG	GENTAG(3)
	TREKANT(L)	FREM(L)
	HØJRE(10)	HØJRE(120)
	HERTIL	HERTIL
	SLUT	SLUT

6.9 Betingede ordrer.

Den betingede ordre giver mulighed til at udføre rekursiv programmering. Syntaksen er følgende

HVIS(<udsagn>)<ordre>

Hvis <udsagn> er sand, udføres <ordre>. Hvis <udsagn> er falsk, overspringes sætningen, og næste programlinje udføres. Hvis <ordre> er SLUT, forlades programmet (niveauet sættes et ned).

<ordre> kan også være et programkald.

Eks. HVIS(N)=0)SLUT
 HVIS(2<=2*N)FIRKANT(L+5)

6.10 Udsagn.

Udsagnet har følgende syntaks:

<udtryk><logisk operator><udtryk>

6.11 Logisk operator.

De logiske operatører er følgende:

1.	=			lig med
2.	>			større end
3.	<			mindre end
4.	>=	eller	=>	større end og lig med
5.	<=	eller	=<	mindre end og lig med
6.	<>	eller	><	forskellig fra

6.12 Programafbrydelse.

Alle programkørsler kan afbrydes ved tryk på en vilkårlig tast. Det er dog ikke muligt at genstarte programmet fra det sted, hvor det blev afbrudt.

7. Indskrivning af programmer.

Skrives ordren HUSK, aktiveres program-editoren, og indskrivningen af programmet kan påbegyndes.

På et hvilket som helst tidspunkt under indskrivningen kan editorens faciliteter benyttes.

- Piltasterne benyttes til at flytte markøren.
- 'DEL' tasten sletter tegnet under markøren og flytter resten af linien en position til venstre.
- 'INS' tasten indsætter et mellemrum under markøren og flytter resten af linien en position til højre.
- 'BS' virker som tasten med pil til venstre.
- Tredie funktionstast fra højre sletter linien under markøren og rykker efterfølgende linier en linie op.
- Anden funktionstast fra højre indsætter en tom linie under markøren og rykker de efterfølgende linier en linie ned.
NB: Hvis der stod noget i nederste linie, mistes det.
- Første funktionstast fra højre sletter alle tegn fra markøren til liniens slutning.

Editoren foretager selv indrykninger/udrykninger af programlinierne.

Programindskrivningen afsluttes automatisk, når 'RETURN' nedtrykkes mens markøren står på linien, hvor ordren SLUT står.

Myresnak-programmer kan maksimalt være på 20 linier. Hver linie kan højst bestå af 30 tegn.

Skrivefeltets bredde er 34 anslag af hensyn til indrykningerne. Når indskrivningen er færdig fjerner editoren automatisk alle mellemrum og afskærer alle tegn udover det 30te.

Programmet lagres i arbejdsområdet, og programnavnet indføres i kataloget. Hvis der allerede er et program med samme navn, gives der mulighed for at rette programnavnet i det sidst indtastede program, ellers overskrives det gamle.

Ordren RET giver også adgang til editoren og benyttes, når man skal foretage rettelser i et eksisterende program.

8. Variabelbegrebet i myresnak.

Myresnak fortolkeren arbejder med to former for variabler nemlig globale og lokale variabler.

Man kan frit skifte mellem de to typer ved ordrene GLOBAL og LOKAL.

Det tilrådes ikke at bruge disse ordre som programordrer, da virkningerne er uoverskuelige.

8.1 Globale variabler.

Ved opstart af myrestart er alle variabler globale, hvilket vises på videosiden i anden linie på skærmens højre halvdel. Alle variablerne er ligeledes nulstillede. (Samme virkning opnås ved ordren OPSTART).

At variablerne er globale betyder, at alle programmerne har adgang til de samme variabler og evt. kan ændre dem.

Ordren NUL nulstiller alle variablerne.

Ordren GLOBAL sætter alle variablerne globale.

Programmerer man rekursivt (dvs. at programmet kalder sig selv) forbliver man på samme niveau (niveau 1), og der er derfor ingen mulighed for tilbagevikling.

Eks. I arbejdsområdet ligger følgende program :

```
TREKANTETSPIRAL(L)
  FREM(L)
  HØJRE(120)
  SPIRAL(L+3)
SLUT
```

Vi starter f.eks. ved ordren TREKANTETSPIRAL(10) på niveau 0 (kommandoniveauet).

Programmet kaldes og tolkes på niveau 1 og L tildeles værdien 10. Myren går 10 frem og højredrejes 120 grader.

Programmet kaldes igen (vi bliver på niveau 1), og L tildeles nu værdien $10+3 = 13$. Myren går 13 frem og højredrejes 120 grader.

Programmet kaldes igen (vi bliver på niveau 1), og L tildeles nu værdien $13+3 = 16$. Myren går 16 frem og højredrejes 120 grader.

Således forsættes i det uendelige, eller til programmet afbrydes ved tryk på en vilkårlig tast. Vi vender så tilbage til kommandoniveauet (niveau 0).

8.2 Lokale variabler.

Variablerne sættes lokale ved ordren LOKAL, hvilket vises på videosiden i anden linie på højre halvdel.

Hvert program får nu sit eget variabelsæt og evt. variabeloverførsler sker ved hjælp af programmets variabeldel. Hvert program har kun adgang til sit eget variabelsæt.

Ved rekursiv programmering vil hver programkald rykke et niveau ud.

Husk: Der er maksimalt 50 niveauer til rådighed.

Ordren NUL nulstiller samtlige variabler på alle niveauer.

Ordren HVIS giver mulighed for tilbagevikling ved rekursive programkald.

Eks. Følgende program ligger i arbejdsområdet :

```
MANGEFIRKANT(L,N)
  HVIS(N=0)SLUT
  GENTAG(4)
    MANGEFIRKANT(L/3,N-1)
    FREM(L)
    VENSTRE(90)
  HERTIL
SLUT
```

Programmet kaldes f.eks. med ordren MANGEFIRKANT(180,3) på kommandoniveauet (niveau 0).

På niveau 1 tildeles variabelen L værdien 180 og N værdien 3. Der testes om N er nul, hvilket den ikke er. Sløjfen påbegynder første gennemløb på niveau 1. Programmet kalder sig selv, og parametrene overføres til niveau 2.

På niveau 2 tildeles variabelen L værdien $180/3 = 60$ og N værdien $3-1 = 2$. Der testes om N er nul, hvilket den ikke er. Sløjfen påbegynder første gennemløb på niveau 2. Programmet kalder sig selv, og parametrene overføres til niveau 3.

På niveau 3 tildeles variabelen L værdien $60/3 = 20$ og N værdien $2-1 = 1$. Der testes om N er nul, hvilket den ikke er. Sløjfen påbegynder første gennemløb på niveau 3. Programmet kalder sig selv, og parametrene overføres til niveau 4.

På niveau 4 tildeles variabelen L værdien $20/3 = 7$ og N værdien $1-1 = 0$. Der testes om N er nul, hvilket den er. Ordren SLUT udføres, og programmet forlader niveau 4. Der hoppes tilbage til niveau 3.

På niveau 3 forsættes med ordren FREM(L). Dvs. myren bevæger sig 20 punkter frem. Tilbageviklingen er altså påbegyndt. Myren venstredrejes 90 grader. Andet gennemløb på

niveau 3 startes, og programmet kalder sig selv.

På niveau 4 får L værdien $20/3 = 7$ og N værdien $1-1 = 0$. Der testes for $N = 0$. Niveaulet forlades påny og vi forsætter på niveau 3.

Således forsættes indtil man når niveau 0.

BEMÆRK at variabelen N benyttes som en slags omvendt niveautæller. Variablen L overføres også til niveau 4, men benyttes ikke.

9. Lidt om fortolkeren.

9.1 Specifikationer.

Fortolkerens tegnsæt består af

- Alle store bogstaver (A til Z).
- Cifrene 0 til 9.
- Parenteser til parameterangivelser.
- Komma til adskillelse af variable/udtryk.
- Minus til angivelse af negative tal.
- De fire regneoperatorer (+, -, * og /).
- De logiske operatorer (>, < og =).

Kommandoliniens længde	30 tegn.
Programliniens længde	30 tegn.
Antal linier pr. program	20 linier.
Antal programmer i arbejdsområdet	20 stk.

Antal programniveauer	50
Antal variable pr. niveau	29

9.2 Fortolkningen.

I korte træk foretages tolkningen af en kommando- eller programlinje som følger :

Tegn til venstre for '(' eller lineslut tolkes, som en ordre.

Hvis ordren ikke findes i fortolkerens ordre-liste, undersøges det, om den findes i programkataloget.

Findes den en af disse steder, udføres ordren ellers gives et signal i højttaleren (dog kun på maskiner med lydfunktion).

Har fortolkeren fundet en '(' søger den efter variable, tal, udtryk og udsagn.

Bruges flere bogstaver i en variabel, benytter fortolkeren kun sidste bogstav som variabelnavn. F.eks. vil variabelen SKR opfattes som en variabel med navnet R.

Udelades skilletegnet ',' mellem variable, opfattes de således kun som en variabel med navnet bestemt af det sidste bogstav i variabelrækken.

Sættes et regnetegn foran et tal eller en variabel, vil kun minus-tegnet tolkes, og tallet/variabelen vil skifte fortegn.

I regneudtryk vil tolkningen ske således :

Eks. Hvis $L=-5$ vil L efter udførelsen af ordren KANT(5*-L) eller KANT(-5*L) have værdien -25 (GLOBAL version).

Hvis der benyttes flere regnetegn sammen (bortset fra minus) benyttes kun det sidst indtastede regnetegn. To minustegn tolkes som eet pludstegn.

Eks. Udtrykket $(K*/+5)$ vil blive tolket som $(K+5)$.

Ved ordren HVIS opfattes alt efter ')' som en ny ordre, som tolkes selvstændigt, hvis udsagnet efter HVIS er sandt.

Eks. Følgende har altså god mening :

HVIS(A=0)HVIS(B=0)SLUT

Kun hvis både A og B er 0 samtidig udføres ordren SLUT.

10. Efterskrift.

Myresnak er et værktøj som andre værktøjer i undervisnings-sammenhænge. Det har een funktion, nemlig at formidle viden og forståelse. Men udnyttelsen af værktøjsfunktionen har mange facetter helt afhængig af hvilke mål og midler brugeren anvender.

Myresnak er et program, der i hele sin ideelle baggrund har et væld af muligheder inden for et utal af fagområder i folkeskolen, gymnasiet etc.

Myresnak er et pædagogisk hjælpemiddel i undervisningen af fremmedsprogede elever.

Myresnak er en pædagogisk vej ved indlæringen af grundbegreber i programmering i faget Datalære.

De ovenstående korte afsnit er påstande, som helt bevidst er udformet polemiske, fordi det er vigtigt, når man arbejder med myresnak, at åbne andre og nye døre og vinduer til pædagogisk metodisk tænkning.

Myresnak er ikke og bør ikke være et etiketteprogram, der kan henføres i de to grupperinger, nemlig

Datamaskin Formidlet Undervisning (DFU)
Datamaskinstyret Undervisning.

Myresnak er til yderste konsekvens et program, der kontrolleres af brugeren i et samarbejde med andre brugere.